1 Introduction

In the next part of the course, we will be working with the Scheme programming language. In addition to learning how to write Scheme programs, we will eventually write a Scheme interpreter in Project 4!

Scheme is a dialect of the Lisp programming language, a language dating back to 1958. The popularity of Scheme within the programming language community stems from its simplicity – in fact, previous versions of CS 61A were taught in the Scheme language.

2 Primitives

Scheme has a set of atomic primitive expressions. Atomic means that these expressions cannot be divided up.

```
(scm> 123)
123

(scm> 123.123)
123.123

(scm> #t)
True

(scm> #f)
False
```

Unlike in Python, the only primitive in Scheme that is a false value is #f and its equivalents, false and False. **This means that 0 is not false.**

The define special form defines variables and procedures by binding a value to a variable, just like the assignment statement in Python. When a variable is defined, the define special form returns a symbol of its name. A procedure is what we call a function in Scheme!

The syntax to define a variable and procedure are:

- `(define <variable name> <value>)`
- `(define (<function name> <parameters>) <function body>)`
3 Call Expressions

Call expressions apply a procedure to some arguments.

\[
\text{(<operator> <operand1> <operand2> \ldots)}
\]

Call expressions in Scheme work exactly like they do in Python. To evaluate them:

1. Evaluate the operator to get a procedure.
2. Evaluate each of the operands from left to right.
3. Apply the value of the operator to the evaluated operands.

For example, consider the call expression \((+ 1 2)\). First, we evaluate the symbol + to get the built-in addition procedure. Then we evaluate the two operands 1 and 2 to get their corresponding atomic values. Finally, we apply the addition procedure to the values 1 and 2 to get the return value 3.

Operators may be symbols, such as + and *, or more complex expressions, as long as they evaluate to procedure values.

```
scm> (- 1 1) ; 1 - 1
0
scm> (/ 8 4 2) ; 8 / 4 / 2
1
scm> (* (+ 1 2) (+ 1 2)) ; (1 + 2) * (1 + 2)
9
```

Some important built-in functions you’ll want to know are:

- +, -, *, /
- =, >, >=, <, <=
- quotient, modulo
- even?, odd?

Questions

3.1 What would Scheme display?

```
scm> (define a (+ 1 2))

a

scm> a

3

scm> (define b (* (* 3 3) (* 4 4)))
```
b

scm> (+ a b)

28

scm> (+ 1 (modulo 10 3) (quotient 10 3))

5

scm> (> (+ 2 3) (* 2 3))

#f

scm> (even? (quotient (* 4 6) 2))

#t

scm> (= (+ (- (* 5 4) 3) 2) 1)

#f

4 Special Forms

Special form expressions contain a special form as the operator. Special form expressions do not follow the same rules of evaluation as call expressions. Each special form has its own rules of evaluation – that’s what makes them special!

The special forms we’ll see in today’s discussion include if, and, or, lambda, and define.

4.1 If Expression

The if special form allows for control flow in our programs. An if expression looks like this:

 \[
 (\text{if} \ \text{<predicate>} \ \text{<if-true>} \ \text{[if-false]})
 \]

<predicate> and <if-true> are required expressions ad [if-false] is optional.

The rules for evaluation are as follows:

1. Evaluate <predicate>.
2. If <predicate> evaluates to a truth-y value, evaluate <if-true> and return its value. Otherwise, evaluate [if-false] if provided and return its value.
This is a special form because not all operands will be evaluated! Only one of the second and third operands is evaluated, depending on the value of the first operand.

Notice that the behavior of an if special form is similar to that of an if statement in Python - the major difference is that an if special form evaluates to a value, whereas if statements, being statements, do not evaluate to anything.

Remember that only #f is a false-y value in Scheme; everything else is truth-y.

```scheme
scm> (if (< 4 5) 1 2)
1
scm> (if #f (/ 1 0) 42)
42
scm> (if (= (+ 1 2) 3) 'equal)
equal
```
4.2 Boolean Operators

Like Python, Scheme also has the boolean operators and, or, and not. and and or are special forms because they are short-circuiting operators.

- **and** takes in any amount of operands and evaluates these operands from left to right until one evaluates to a false-y value. It either returns that first false-y value or the value of the last expression if all operands evaluate to a truth-y value.

- **or** takes in any amount of operands and evaluates these operands from left to right until one evaluates to a truth-y value. It either returns that first truth-y value or the value of the last expression if all operands evaluate to a false-y value.

- **not** takes in a single operand, evaluates it, and returns its opposite truthiness value. Because its one operand is always evaluated, not is a regular procedure rather than a special form.

```scheme
scm> (and 25 32)
32
scm> (or #f (< 4 3) (= 1 2))
#f
scm> (or 1 (/ 1 0)) ; Short-circuits
1
scm> (not (odd? 10)) ; Not a special form!
#t
```

Questions

4.1 What would Scheme display?

```scheme
scm> (if (or #t (/ 1 0)) 1 (/ 1 0))
1

scm> (if (> 4 3) (+ 1 2 3 4) (+ 3 4 (* 3 2)))
10

scm> ((if (< 4 3) +) 4 100)
-96

scm> (if 0 1 2)
1
```
4.3 Lambdas and Defining Functions

All Scheme procedures are lambda procedures. One way to create a procedure is to use the `lambda` special form.

```
(lambda (<param1> <param2> ...) <body>)
```

This expression creates a lambda function with the given parameters and body, but does not evaluate the body. Just like in Python, the body expression is not evaluated until the lambda function is called and applied to some argument values. The fact that neither operand is evaluated is what makes `lambda` a special form.

Another similarity to Python is that lambda expressions do not assign the returned function to any name. We can assign the value of an expression to a name with the `define` special form:

```
(define <name> <expr>)
```

For example, `(define square (lambda (x) (* x x)))` creates a lambda procedure that squares its argument and assigns that procedure to the name `square`.

The second version of the `define` special form is a shorthand for this function definition:

```
(define (<name> <param1> <param2 ...>) <body>)
```

This expression creates a function with the given parameters and body and binds it to the given name.

```
scm> (define square (lambda (x) (* x x))) ; Bind the lambda function to the name square
square
scm> (define (square x) (* x x)) ; Equivalent to the line above
square
scm> square
(lambda (x) (* x x))
scm> (square 4)
16
```

Questions

4.1 Write a function that returns the factorial of a number.

```
(define (factorial x)
  (if (< x 2)
      1
      (* x (factorial (- x 1)))))
```

4.2 Write a function that returns the $n^{th}$ Fibonacci number.

```
(define (fib n)
```


(if (or (= n 0) (= n 1))
    n
    (+ (fib (- n 1)) (fib (- n 2)))))
5 Pairs and Lists

Scheme lists implement an abstract data type known as a linked list with a sequence of pairs. We define a list to be one of two things:

- the empty list, nil
- a pair whose second element is a list

Linked lists are recursive data structures. The base case is the empty list.

We can use the same procedures used to construct and select from pairs to construct and select from lists:

- \( \text{cons first rest} \) constructs a list with the given first element and rest of the list
- \( \text{car lst} \) gets the first item of the list
- \( \text{cdr lst} \) gets the rest of the list

```
scm> nil
()
scm> (define lst (cons 1 (cons 2 (cons 3 nil))))
lst
scm> lst
(1 2 3)
scm> (car lst)
1
scm> (cdr lst)
(2 3)
scm> (car (cdr lst))
2
scm> (cdr (cdr (cdr lst)))
()
```

In general, the rule for displaying a pair is as follows: a dot separates the \text{car} and \text{cdr} fields of a pair. However, if the dot is immediately followed by an open parenthesis, then remove the dot and the parenthesis pair. Thus, \((\emptyset . (1 . 2))\) becomes \((\emptyset 1 . 2)\). Intuitively, a dot will be followed by an open parenthesis if the \text{cdr} of a pair is another pair.

```
scm> (cons 1 (cons 2 3))
(1 2 . 3)
scm> (cons 1 (cons 2 (cons 3 nil)))
(1 2 3)
```

Two other common ways of creating lists is using the built-in \text{list} procedure or the \text{quote} special form:

- The \text{list} procedure takes in an arbitrary amount of arguments. Because it is a procedure, all operands are evaluated when \text{list} is called. A list is constructed with the values of these operands and is returned.
• The quote special form takes in a single operand. It returns this operand exactly as is, without evaluating it. Note that this special form can be used to return any value, not just a list.

```
scm> (list 1 2 3)
(1 2 3)
scm> (quote (1 2 3))
(1 2 3)
scm> '(1 2 3) ; Equivalent to the previous quote expression
(1 2 3)
```

To check if a list is a well-formed list (i.e. it fits one of the descriptions in the definition above), use the predicate procedure `list?`.

```
=, eq?, equal?
```

• `=` can only be used for comparing numbers.

• `eq?` behaves like `==` in Python for comparing two non-pairs (numbers, booleans, etc.). Otherwise, `eq?` behaves like `is` in Python.

• `equal?` compares pairs by determining if their cars are `equal?` and their cdrs are `equal?`(that is, they have the same contents). Otherwise, `equal?` behaves like `eq?`.

```
scm> (define a '(1 2 3))
a
scm> (= a a)
Error
scm> (equal? a '(1 2 3))
#t
scm> (eq? a '(1 2 3))
#f
scm> (define b a)
b
scm> (eq? a b)
#t
```
Questions

5.1 Write a function which takes two lists and concatenates them.

Notice that simply calling `(cons a b)` would not work because it will create a deep list.

```scheme
(define (concat a b)
  (if (null? a)
      b
      (cons (car a) (concat (cdr a) b))))
```

**Video walkthrough**

```scheme
scm> (concat '(1 2 3) '(2 3 4))
(1 2 3 2 3 4)
```

5.2 Write a function that takes an element `x` and a non-negative integer `n`, and returns a list with `x` repeated `n` times.

```scheme
(define (replicate x n)
  (if (= n 0)
      nil
      (cons x (replicate x (- n 1)))))
```

**Video walkthrough**

```scheme
scm> (replicate 5 3)
(5 5 5)
```
5.3 A run-length encoding is a method of compressing a sequence of letters. The list
(a a a b a a a a) can be compressed to ((a 3) (b 1) (a 4)), where the compressed
version of the sequence keeps track of how many letters appear consecutively.

Write a function that takes a compressed sequence and expands it into the original
sequence. **Hint:** You may want to use `concat` and `replicate`.

```scheme
(define (uncompress s)
  (if (null? s)
    s
    (concat (replicate (car (car s)) (car (cdr (car s))))
            (uncompress (cdr s))))
)
```

**Video walkthrough**

```scheme
scm> (uncompress '((a 1) (b 2) (c 3)))
(a b b c c c)
```

5.4 Write a function that takes a procedure and applies it to every element in a given
list.

```scheme
(define (map fn lst)
  (if (null? lst)
    nil
    (cons (fn (car lst)) (map fn (cdr lst))))
)
```

```scheme
scm> (map (lambda (x) (* x x)) '(1 2 3))
(1 4 9)
```

5.5 Write a function that takes a procedure and applies to every element in a given
nested list.

The result should be a nested list with the same structure as the input list, but
with each element replaced by the result of applying the procedure to that element.

Use the built-in `list?` procedure to detect whether a value is a list.

```scheme
(define (deep-map fn lst)
  (cond ((null? lst) lst)
        ((list? (car lst)) (cons (deep-map fn (car lst)) (deep-map fn (cdr lst))))
        (else (cons (fn (car lst)) (deep-map fn (cdr lst))))
  )
)
```

```scheme
scm> (deep-map (lambda (x) (* x x)) '(1 2 3))
(1 4 9)
scm> (deep-map (lambda (x) (* x x)) '(1 ((4) 5) 9))
(1 ((16) 25) 81)
```
6 Extra Questions

6.1 Fill in the following to complete an abstract tree data type:

(define (make-tree label branches) (cons label branches))

(define (label tree)
  (car tree))

(define (branches tree)
  (cdr tree))

6.2 Using the abstract data type above, write a function that sums up the entries of a tree, assuming that the entries are all numbers.

Hint: you may want to use the map function you defined above, and also write a helper function for summing up the entries of a list.

(define (tree-sum tree)
  (+ (label tree) (sum (map tree-sum (branches tree))))))

(define (sum lst)
  (if (null? lst) 0 (+ (car lst) (sum (cdr lst))))))

6.3 Using the abstract data type above, write a function that creates a new tree where the entries are the product of the entries along the path to the root in the original tree.

Hint: you may want to write a helper function that keeps track of the current product.

(define (path-product-tree t)
  (define (path-product t product)
    (let ((prod (* product (label t))))
      (make-tree prod
        (map (lambda (t) (path-product t prod))
          (branches tree))))))
  (path-product t 1))